Changes in F-value of instrument I am using 2SLS to estimate the effect of education on the probability that one works. In the first stage I regress education on my instrument and the other exogenous control variables. The same exogenous control variables are then included in the second stage.
In my main estimations I have 8 countries. As a test of the robustness of the instrument I exclude countries and investigate if the first-stage results remain. The results are fairly similar when excluding most of the countries. However, when one of the countries is excluded the F-statistics of the instrument falls and basically becomes 0. When removing another country instead, the F-statistics increases threefold. What could these great changes in the F-statistics possibly depend on?  
 A: After having bothered you with all my follow up and clarification questions the most likely reason for this to happen is that education in one country is less correlated with changes in minimum schooling laws than another. This changes the strength of your instruments across countries. Reasons for such behavior can be cultural, institutional or whatever. The same policy in two different countries may well produce two different effects.
The other reason might be the sample size because having 3000 observations versus 6000 observations in the single samples implies different magnitudes of power loss when moving away from the full sample. For the smaller sample you would also expect larger t-statistics and confidence intervals simply because you use much fewer individuals.
To illustrate this I have written up a very simple simulation exercise. If you don't use Stata, that's okay because everything is annotated. I have created 4 countries with 2 time periods with 2000 individuals each (you might think of those time periods as two different birth cohorts in each country). In the second period the minimum school leaving age increases such that the second cohort has higher values of schooling.
// Set a seed such that we can reproduce the results later
set seed 1234

// create 8 observations and number them from 1 to 8, and generate the time period t = {1,2}
// obs 1-4 will by country 1-4 in period 1, obs 5-8 will be country 1-4 in period 2
set obs 8
gen c = _n
gen t = (c>4) + 1

// generate the instrument for each country in each period
// the average of school leaving age is 8 in t=1, and 11 in t=2
// I assign them randomly and then round the numbers
gen Z = rnormal(11,0.5) if t==2
replace Z = rnormal(8, 0.5) if t==1
replace Z = round(Z)

// now generate 1000 individuals in each period
// recode the county id
expand 1000
recode c (5 = 1) (6 = 2) (7 = 3) (8 = 4)

// generate the regression error
gen e = rnormal(0,0.5)

// generate the education variable which depends on the min school leaving age Z and the error (to make it endogenous)
// S will depend more on Z in country 1, less in country 4
// S will have less noise in country 1, more in country 4
gen S = 4 + 1.2*Z + 0.1*e if c==1
replace S = 4 + 0.8*Z + e if c==2 | c==3
replace S = 4 + 0.4*Z + 2*e if c==4

replace S = round(S)

bysort t: sum S
-> t = 1

    Variable |       Obs        Mean    Std. Dev.       Min        Max
-------------+--------------------------------------------------------
           S |      4000    10.49125    2.487865          4         14

-> t = 2

    Variable |       Obs        Mean    Std. Dev.       Min        Max
-------------+--------------------------------------------------------
           S |      4000     12.7495    3.119003          5         17


// generate the outcome variable (e.g. earnings) with the true coeff for S being 0.1
gen y = 1.2 + 0.1*S + e

All of this was a bit lengthy to explain the data set up, the rest will be quick. If you run a 2SLS regression, instrumenting S with Z, you get a first stage t-statistic of 35.80 using the whole sample. I don't report the whole result table for brevity but the code for this is:
ivreg y (S=Z), first

If you repeat the 2SLS regression for each country separately,
bysort c: ivreg y (S=Z), first

you get:
country     first stage t-stat
1           698.93
2            94.16
3            92.48
4            24.41

which is somewhat smaller for country 4 but much much larger for country 1. Now if we half the sample of country 4 by randomly dropping people,
gen x = rnormal(0,1) if c==4
drop if x<0
ivreg2 y (S=Z) if c==4, first

you get a t-statistic for the instrument of 17.66 which is 25% lower of what you had before.
To sum up, the reasons for your result can be:


*

*more noise in one country

*a stronger effect of the law change in one country

*the smaller sample size

